
1 [bookmark: _Toc122434485][bookmark: _Toc112921493]Overview
This document lists all the essencial changes needed to correct problems in the TTCN implementation of NR5GC test case 11.4.12 which is part of the NR5GC test suite in iWD_TTCN3-B2020-09_D22wk23.
With these changes applied the test case can be demonstrated to run with one or more UEs (see section 4). Execution log file(s) is (are) provided as evidence. 
Contact:	Parikshit Bhise
	parikshit.bhise@rohde-schwarz.com 
	Shaun Harry
	shaun.harry@keysight.com
	
1. [bookmark: _Toc112921494]Verification Test Summary 
Test Case: 	11.4.12.NR5GC
ATS Version:	iwd-TTCN3-B2020-09_D22wk23
[bookmark: _Hlk37919671]System Simulator used:	R&S® 5G Protocol Conformance Test platform
UE used:	Qualcomm SM 8450,SDX65 UE  
Verification Status:	PASS



2 [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc112921495]Corrections required to test case 11.4.12
2.1 [bookmark: _Toc112921496][bookmark: _Toc122434493][bookmark: _Toc295288970][bookmark: _Toc325725665]f_TC_11_4_12_NR5GC
	Function name
	f_TC_11_4_12_NR5GC

	Reason for change
	In the current TTCN, The NR Cell is set as serving cell power before the EUTRA Registration is done. This results in UE finding the NR cell ahead of EUTRA Cell and UE tries to camp on NR Cell and fails the test case in the preamble.

This needs to be addressed.

	Summary of change
	It is proposed to set NR Cell 1 to “Serving cell” power after EUTRA registration.


	TTCN module
	Emergency_IRAT_NR5GC_NR. ttcn

	MCC160 Comment
	Accepted



Before Change:
	function f_TC_11_4_12_NR5GC() runs on NR5GC_PTC
  { //5GMM-REGISTERED.NORMAL-SERVICE / 5GMM-IDLE / Emergency call / Disabling N1 mode / Emergency call establishment over EPS / Success
    var NAS_PlmnId v_PLMN_cell;
    var TrackingAreaCode v_TAC_cell;
    var template (value) NG_TrackingAreaIdList v_TAIList;
    var template (value) NG_NAS_DL_Message_Type v_MsgToSend;
    var GMM_MobilityInfo_Type v_GMM_MobilityInfo;

    f_NR5GC_IRAT_Init(NR_6);

    //Create and configure cell
    f_NR_CellConfig_Def(nr_Cell1);
    f_NR_SetCellPower (nr_Cell1, -88, -82);

    // First pass capability info to EUTRA
    f_NR5GC_IRAT_UECapabilityTransfer(nr_Cell1);
    // Wait until UE has attached on EUTRA
    f_EUTRA_NR_WaitForCoOrd_Trigger (EUTRA, "Preamble");
<<SKIPPED CODE>>



After Change:
	function f_TC_11_4_12_NR5GC() runs on NR5GC_PTC
  { //5GMM-REGISTERED.NORMAL-SERVICE / 5GMM-IDLE / Emergency call / Disabling N1 mode / Emergency call establishment over EPS / Success
    var NAS_PlmnId v_PLMN_cell;
    var TrackingAreaCode v_TAC_cell;
    var template (value) NG_TrackingAreaIdList v_TAIList;
    var template (value) NG_NAS_DL_Message_Type v_MsgToSend;
    var GMM_MobilityInfo_Type v_GMM_MobilityInfo;

    f_NR5GC_IRAT_Init(NR_6);

    //Create and configure cell
    f_NR_CellConfig_Def(nr_Cell1);
    //Commented out f_NR_SetCellPower (nr_Cell1, -88, -82); //WA#11.4.12 WA#WI997012

    // First pass capability info to EUTRA
    f_NR5GC_IRAT_UECapabilityTransfer(nr_Cell1);
    // Wait until UE has attached on EUTRA
    f_EUTRA_NR_WaitForCoOrd_Trigger (EUTRA, "Preamble");
    f_NR_SetCellPower (nr_Cell1, -88, -82); //WA#11.4.12 WA#WI997012

<<SKIPPED CODE>>



2.2 [bookmark: _Toc112921497][bookmark: _Toc112921498]fl_TC_11_4_12_NR5GC_EUTRA_TestBody
	Function name
	fl_TC_11_4_12_NR5GC_EUTRA_TestBody

	Reason for change
	· In the current TTCN, there is a specific check for ePCO bit in TRACKING AREA UPDATE REQUEST message is sent to update the registration of the actual tracking area, but as per prose, this test case title itself mentions that N1 mode is disabled. Hence checking for IE ePCO to be set to 1 is incorrect. Similar changes have been made and accepted for EPS fallback testacse 11.1.7

· In the current TTCN, there is no coordination message present to tell IMS PCT that RAT has changed after TRACKING AREA UPDATE COMPLETE. This change is needed because of Change 2.3

· In the current TTCN once the call is established, there is no co-ordination message from Eutra PTC to IMS PTC that informs IMS PTC to release the call. 

· In the Current TTCN there is no handling for “Release IMS Call as specified in the generic procedure in TS 34.229-1 [35] subclause C.32.”. Basically the procedure to Deactivate the EPS Bearer needs to be implemented.



	Summary of change
	· Removed the check for ePCO bit at step 4. Prose CR is also required

· Called function f_IMS_IPCAN_SendCoOrdMsg (IMS [tsc_Index_IMS1] in function fl_TC_11_4_12_NR5GC_EUTRA_TestBody after call to f_EUTRA38_TrackingAreaUpdateFromN1_WithoutReq to inform IMS PTC the RAT has changed. This is necessary because of change 2.3

· Called function f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_IMS2]) in function fl_TC_11_4_12_NR5GC_EUTRA_TestBody() after call to f_EUTRA38_MultiPDN_ActivateDedicatedEpsBearer_Speech(eutra_Cell1, Emergency_PDN) to tell IMS PTC to release the call. 

· Called function f_EUTRA_DeactivateEPS_BearerContext (eutra_Cell1,                                int2hex(valueof(f_Get_EPSBearerIdForAPNType(Emergency_PDN)), 1)) after the coordination message to release the call is received from IMS PTC in function fl_TC_11_4_12_NR5GC_EUTRA_TestBody.



	TTCN module
	Emergency_IRAT_NR5GC_EUTRA.ttcn

	MCC160 Comment
	1. Accepted and associated prose CR R5-226306 was agreed at RAN5#97
R5-226306
2.  Not needed, according to the TF160 comment of Change 2.3. 
3. Accepted
4. Accepted, however it has been implemented based on TTCN CR R5s211357




Before Change:
	function fl_TC_11_4_12_NR5GC_EUTRA_TestBody() runs on EUTRA_Multi5GS_PTC
  {

    var NAS_MSG_Indication_Type v_NasInd;
    var NasCount_Type v_NasCount;

    //@siclog "Step 0" siclog@ //@sic R5-223438 sic@
    //Set the power levels according to T0 as per Table 11.4.12.3.1-1/2
    f_EUTRA38_SetCellPower(eutra_Cell1, -91, -91);

    //@siclog "Step 1" siclog@
    //Make the UE initiate an Emergency call.
    f_UT_RequestIMSEmergencyCall(UT);

    //@siclog "Step 2 - 4" siclog@
    //Check: Does the UE transmit an RRCConnectionRequest message with 'establishmentCause' set to 'emergency'?
    //SS transmits an RRCConnectionSetup message.
    //Check: Does the UE transmit an RRCConnectionSetupComplete message to confirm the successful completion of the connection establishment and a TRACKING AREA UPDATE REQUEST message is sent to update the registration of the actual tracking area.
    //For some consequences related to the content of the TRACKING AREA UPDATE REQUEST message see the Notes in TS.36.508 [7], Table 4.5A.2.1-1.
    v_NasInd := f_EUTRA38_TAUReqFromN1(eutra_Cell1,
                                       tsc_RRC_TI_Def,
                                       -,
                                       MAPPED_CXT,
                                       -,
                                       emergency);
                                       
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 2");

    if (not match (v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.ueNetworkCapability, cdr_NR5GC_UENetworkCap_11_4_12('58'O))){
      f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "ueNetworkCapability indication is not correct at step 4");
    }
    else {
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 4");
    };

    v_NasCount := v_NasInd.SecurityProtection.NasCount;
    
    //@siclog "Steps 5-8" siclog@
    //Steps 5-8 from the Generic Test Procedure for IMS Emergency call establishment in EUTRA: Normal Service as specified in TS 36.508 [7], Table 4.5A.4.3-1 are performed.
    f_EUTRA38_MultiPDN_Activate_SRB2_DRB_SendRrcMsg(eutra_Cell1,
                                         tsc_RRC_TI_Def,
                                         v_NasCount);

    f_EUTRA_Activate_SRB2_DRB_ReceiveRrcAndNasMsg(eutra_Cell1,
                                                  tsc_RRC_TI_Def);

    //@siclog "Step 9" siclog@
    //SS responds with TRACKING AREA UPDATE ACCEPT message.
    //@siclog "Step 10" siclog@
    //The UE transmits a TRACKING AREA UPDATE COMPLETE message.
    f_EUTRA38_TrackingAreaUpdateFromN1_WithoutReq(eutra_Cell1,
                                                    v_NasInd,
                                                    MAPPED_CXT,
                                                    noRrcConnectionRelease,
                                                    -, true,
                                                    -, -, true);
                                                                
    //@siclog "Steps 11-17" siclog@
    //Steps 9-15 from the Generic Test Procedure for IMS Emergency call establishment in EUTRA: Normal Service as specified in TS 36.508 [7], Table 4.5A.4.3-1 are performed.
    f_EUTRA38_MultiPDN_EstablishAdditionalPDNs(eutra_Cell1, 1, -, -, -, '100'B, omit); // @sic R5s211357 sic@

    // @siclog "Step 19-20" siclog@
    f_EUTRA38_MultiPDN_ActivateDedicatedEpsBearer_Speech(eutra_Cell1, Emergency_PDN);

    // Emergency call release in IMS
    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_IMS2]);

  }
}



After Change:
	function fl_TC_11_4_12_NR5GC_EUTRA_TestBody() runs on EUTRA_Multi5GS_PTC
  {

    var NAS_MSG_Indication_Type v_NasInd;
    var NasCount_Type v_NasCount;

    //@siclog "Step 0" siclog@ //@sic R5-223438 sic@
    //Set the power levels according to T0 as per Table 11.4.12.3.1-1/2
    f_EUTRA38_SetCellPower(eutra_Cell1, -91, -91);

    //@siclog "Step 1" siclog@
    //Make the UE initiate an Emergency call.
    f_UT_RequestIMSEmergencyCall(UT);

    //@siclog "Step 2 - 4" siclog@
    //Check: Does the UE transmit an RRCConnectionRequest message with 'establishmentCause' set to 'emergency'?
    //SS transmits an RRCConnectionSetup message.
    //Check: Does the UE transmit an RRCConnectionSetupComplete message to confirm the successful completion of the connection establishment and a TRACKING AREA UPDATE REQUEST message is sent to update the registration of the actual tracking area.
    //For some consequences related to the content of the TRACKING AREA UPDATE REQUEST message see the Notes in TS.36.508 [7], Table 4.5A.2.1-1.
    v_NasInd := f_EUTRA38_TAUReqFromN1(eutra_Cell1,
                                       tsc_RRC_TI_Def,
                                       -,
                                       MAPPED_CXT,
                                       -,
                                       emergency);
                                       
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 2");

    if (not match (v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.ueNetworkCapability, cdr_NR5GC_UENetworkCap_11_4_12('58'O))){
      f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "ueNetworkCapability indication is not correct at step 4");
    }
    else {
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 4");
    };
      
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 4"); 
      
    v_NasCount := v_NasInd.SecurityProtection.NasCount;
    
    //@siclog "Steps 5-8" siclog@
    //Steps 5-8 from the Generic Test Procedure for IMS Emergency call establishment in EUTRA: Normal Service as specified in TS 36.508 [7], Table 4.5A.4.3-1 are performed.
    f_EUTRA38_MultiPDN_Activate_SRB2_DRB_SendRrcMsg(eutra_Cell1,
                                         tsc_RRC_TI_Def,
                                         v_NasCount);

    f_EUTRA_Activate_SRB2_DRB_ReceiveRrcAndNasMsg(eutra_Cell1,
                                                  tsc_RRC_TI_Def);

    //@siclog "Step 9" siclog@
    //SS responds with TRACKING AREA UPDATE ACCEPT message.
    //@siclog "Step 10" siclog@
    //The UE transmits a TRACKING AREA UPDATE COMPLETE message.
    f_EUTRA38_TrackingAreaUpdateFromN1_WithoutReq(eutra_Cell1,
                                                    v_NasInd,
                                                    MAPPED_CXT,
                                                    noRrcConnectionRelease,
                                                    -, true,
                                                    -, -, true);
    // Tell IMS PTC the RAT has changed
    f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_IMS1], cms_IPCAN_IMS_QueryResponse(f_EUTRA_IMS_IpcanInfo()));//WA#11.4.12 WA#WI997012
                                                                
    //@siclog "Steps 11-17" siclog@
    //Steps 9-15 from the Generic Test Procedure for IMS Emergency call establishment in EUTRA: Normal Service as specified in TS 36.508 [7], Table 4.5A.4.3-1 are performed.
    f_EUTRA38_MultiPDN_EstablishAdditionalPDNs(eutra_Cell1, 1, -, -, -, '100'B, omit); // @sic R5s211357 sic@

    // @siclog "Step 19-20" siclog@
    f_EUTRA38_MultiPDN_ActivateDedicatedEpsBearer_Speech(eutra_Cell1, Emergency_PDN);
         
     // Tell IMS PTC to release the call
    f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_IMS2]);//WA#11.4.12 WA#WI997012

    // Emergency call release in IMS
    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_IMS2]);
    
   // The SS deactivates the EPS bearer. [This is the rest of C.32]
    f_EUTRA_DeactivateEPS_BearerContext (eutra_Cell1,
                                         int2hex(valueof(f_Get_EPSBearerIdForAPNType(Emergency_PDN)), 1));//WA#11.4.12 WA#WI997012

  }
}




2.3 f_TC_11_4_12_NR5GC_IMS1
	Function name
	f_TC_11_4_12_NR5GC_IMS1

	Reason for change
	In the current TTCN implementation, the IMS default handler will be used to handle the IMS Emergency Registration. This causes problems during the Emergency call setup. 
It is proposed to introduce a new function f_TC_11_4_12_NR5GC_IMS1, which will handle the emergency registration and avoid using the default handler. This mechanism is followed in EPS fallback testcase 11.1.7 as well. 

	Summary of change
	New Function f_TC_11_4_12_NR5GC_IMS1 added. 

	TTCN module
	Emergency_IRAT_NR5GC_IMS.ttcn

	MCC160 Comment
	Not needed
The reason for change seems not aligned with the proposed code. IMS1 PTC is used to handle normal IMS registration in Preamble and optional IMS deregistration for the normal IMS PDN, not for the emergency IMS PDN. Furthermore it is expected that the default handler should be capable of handling normal IMS registration/deregistration for the normal IMS PDN.




	
module Emergency_IRAT_NR5GC_IMS {
 
  import from IMS_Component all;
  import from LibSip_SIPTypesAndValues all;
  import from IMS_PTC_CoordMsg all;
  import from IMS_CommonFunctions all;
  import from IMS_Procedures_Common all;
  import from IMS_Procedures_Registration all;
  import from IMS_ASP_TypeDefs all;
  import from IMS_ASP_Templates all;
  import from IMS_SIP_Templates all;
  import from IMS_NR5GC_Functions all;
  import from IMS_Procedures_CallControl4G5G all;
  import from CommonDefs all;
  import from EPS_Fallback_NR5GC_IMS all;//WA#11.4.12 WA#WI997012



function f_TC_11_4_12_NR5GC_IMS1()runs on IMS_PTC//WA#11.4.12 WA#WI997012:New Function
  { 
    var IMS_IPCAN_Coordination_MSG v_IMS_IPCAN_CoordMsg;
    
    fl_IMSFallback_PreambleRegistrations();
    
    // Wait for EUTRA to say that the RAT has changed
    OtherIPCAN.receive(cmr_IPCAN_IMS_Response) -> value v_IMS_IPCAN_CoordMsg;
    f_IMS_PTC_SetRanType(v_IMS_IPCAN_CoordMsg.IPCAN_IMS_Response.IpcanInfo.RanType);
    
    f_IMS_IpcanReleaseWithOptionalDeregistrationOtherIPCAN();
  }



2.4 [bookmark: _Toc112921499]TC_11_4_12_NR5GC
	Function name
	TC_11_4_12_NR5GC

	Reason for change
	As proposed in change 2.3, a new function is defined to handle the IMS emergency Registration. This needs to be now accommodated in the main function TC_11_4_12_NR5GC, and corresponding variables be initialized. 

	Summary of change
	v_IMS1 := IMS_PTC.create("IMS1") alive;  and v_IMS1.start(f_TC_11_4_12_NR5GC_IMS1()); are added. 

	TTCN module
	NR5GC_IRAT_Testsuite.ttcn

	MCC160 Comment
	Not needed according to the TF160 comment for Change 2.3




Before Change:
	testcase TC_11_4_12_NR5GC() runs on MTC_NR5GC_IRAT system SYSTEM_NR5GC_IRAT {
    // @purpose
    //   5GMM-REGISTERED.NORMAL-SERVICE / 5GMM-IDLE / Emergency call / Disabling N1 mode / Emergency call establishment over EPS / Success
    var NR5GC_PTC        v_NR5GC      := null;
    var EUTRA_Multi5GS_PTC v_EUTRA_Multi5GS := null;
    var UTRAN_NR5GC_PTC  v_UTRAN_NR5GC := null;
    var IMS_PTC          v_IMS1       := null;
    var IMS_PTC          v_IMS2       := null;
    var WLAN_PTC         v_WLAN       := null;
  
    timer t_GuardTimer := int2float(300);
  
    v_NR5GC := NR5GC_PTC.create("NR5GC") alive;
    v_EUTRA_Multi5GS := EUTRA_Multi5GS_PTC.create("EUTRA_Multi5GS") alive;
    v_IMS2 := IMS_PTC.create("IMS2") alive;
  
    f_MTC_ConnectPTCs_NR5GC_IRAT(system, v_NR5GC, v_EUTRA_Multi5GS, v_UTRAN_NR5GC, v_IMS1, v_IMS2, v_WLAN);
  
    v_NR5GC.start(f_TC_11_4_12_NR5GC());
    v_EUTRA_Multi5GS.start(f_TC_11_4_12_NR5GC_EUTRA());
    v_IMS2.start(f_TC_11_4_12_NR5GC_IMS2());
  
    t_GuardTimer.start;
  
    f_MTC_MainLoop_NR5GC_IRAT(t_GuardTimer);
  }






After Change:
	testcase TC_11_4_12_NR5GC() runs on MTC_NR5GC_IRAT system SYSTEM_NR5GC_IRAT {
    // @purpose
    //   5GMM-REGISTERED.NORMAL-SERVICE / 5GMM-IDLE / Emergency call / Disabling N1 mode / Emergency call establishment over EPS / Success
    var NR5GC_PTC        v_NR5GC      := null;
    var EUTRA_Multi5GS_PTC v_EUTRA_Multi5GS := null;
    var UTRAN_NR5GC_PTC  v_UTRAN_NR5GC := null;
    var IMS_PTC          v_IMS1       := null;
    var IMS_PTC          v_IMS2       := null;
    var WLAN_PTC         v_WLAN       := null;
  
    timer t_GuardTimer := int2float(300);
  
    v_NR5GC := NR5GC_PTC.create("NR5GC") alive;
    v_EUTRA_Multi5GS := EUTRA_Multi5GS_PTC.create("EUTRA_Multi5GS") alive;
    v_IMS1 := IMS_PTC.create("IMS1") alive;//WA#11.4.12 WA#WI997012
    v_IMS2 := IMS_PTC.create("IMS2") alive;
  
    f_MTC_ConnectPTCs_NR5GC_IRAT(system, v_NR5GC, v_EUTRA_Multi5GS, v_UTRAN_NR5GC, v_IMS1, v_IMS2, v_WLAN);
  
    v_NR5GC.start(f_TC_11_4_12_NR5GC());
    v_EUTRA_Multi5GS.start(f_TC_11_4_12_NR5GC_EUTRA());
    v_IMS1.start(f_TC_11_4_12_NR5GC_IMS1()); //WA#11.4.12 WA#WI997012
    v_IMS2.start(f_TC_11_4_12_NR5GC_IMS2());
  
    t_GuardTimer.start;
  
    f_MTC_MainLoop_NR5GC_IRAT(t_GuardTimer);
  }


2.5 [bookmark: _Toc112921500]f_TC_11_4_12_NR5GC_IMS2
	Function name
	f_TC_11_4_12_NR5GC_IMS2

	Reason for change
	· In the current TTCN, the IMS PTC is not waiting for a trigger from IPCAN when IMS call is released 

· In the current TTCN, p_AutonomousCallRelease is set to true for f_IMS_EmergencyCallRelease which is incorrect and due to this there is no MMI prompt to hangup the call.

	Summary of change
	· f_IMS_IPCAN_WaitForTrigger(OtherIPCAN) is added at step 19. 

· Change the value to false for p_AutonomousCallRelease in the function 
           f_IMS_EmergencyCallRelease(v_InviteRequestEmergencyCall,false /*true*/);


	TTCN module
	Emergency_IRAT_NR5GC_IMS.ttcn

	MCC160 Comment
	1. Accepted
2. Accepted




Before Change:
	   function f_TC_11_4_12_NR5GC_IMS2()runs on IMS_PTC
  { // Emergency PDU session

    var INVITE_Request v_InviteRequestEmergencyCall;


    f_IMS_PTC_Init(PDN_2);

    // Steps 1 to 4 of the generic procedure for IMS emergency registration (TS 34.229-1 C.20).
    f_IMS_EmergencyRegistration(-, -, OtherIPCAN_E);
    // Steps 1 to 5 of the generic procedure for IMS emergency speech call establishment (TS 34.229-1 C.22).
    v_InviteRequestEmergencyCall := f_IMS_EmergencyCallSetup();

    f_Delay( 1.0 );

    //@siclog "Step 19" siclog@
    //Release IMS Call as specified in the generic procedure in TS 34.229-1 [35] subclause C.32.
    f_IMS_EmergencyCallRelease(v_InviteRequestEmergencyCall, true);

    // Inform NR: Emergency Call released.
    f_IMS_IPCAN_SendCoOrdMsg(OtherIPCAN);

  }
}





After Change:
	function f_TC_11_4_12_NR5GC_IMS2()runs on IMS_PTC
  { // Emergency PDU session

    var INVITE_Request v_InviteRequestEmergencyCall;


    f_IMS_PTC_Init(PDN_2);

    // Steps 1 to 4 of the generic procedure for IMS emergency registration (TS 34.229-1 C.20).
    f_IMS_EmergencyRegistration(-, -, OtherIPCAN_E);
    // Steps 1 to 5 of the generic procedure for IMS emergency speech call establishment (TS 34.229-1 C.22).
    v_InviteRequestEmergencyCall := f_IMS_EmergencyCallSetup();

    f_Delay( 1.0 );
      
    //@siclog "Step 19" siclog@
    //Release IMS Call as specified in the generic procedure in TS 34.229-1 [35] subclause C.32.

    f_IMS_IPCAN_WaitForTrigger(OtherIPCAN);//WA#11.4.12 WA#WI997012
   
    f_IMS_EmergencyCallRelease(v_InviteRequestEmergencyCall,false /* true WA#11.4.12 WA#WI*/);

    // Inform NR: Emergency Call released.
    f_IMS_IPCAN_SendCoOrdMsg(OtherIPCAN);

  }
}




2.6 [bookmark: _Toc112921501]cdr_NR5GC_UENetworkCap_11_4_12
	Function name
	cdr_NR5GC_UENetworkCap_11_4_12

	Reason for change
	The conformance requirement 24.301 5.5.3.2.2 states “If the UE supports NB-S1 mode, Non-IP PDN type, or N1 mode, then the UE shall support the extended protocol configuration options IE." At step 4, the UE has disabled support of N1 mode is disabled so it may not indicate support for ePCO any more.

Delete specific checking for octet 8 bit 8 (ePCO support) in UE network capability at step 4.

	Summary of change
	Remove checking of ePCO support bit 
Note : An associated prose CR on 38.523-1 will be raised at RAN5#97

	TTCN module
	Emergency_IRAT_NR5GC_EUTRA

	MCC160 Comment
	Accepted and associated prose CR R5-226306 was agreed at RAN5#97
R5-226306





Before Change:
	template (present) UE_NetworkCap cdr_NR5GC_UENetworkCap_11_4_12 (template (omit) IEI8_Type p_IEI)
           modifies cr_UENetworkCap_Any :=
  {
    cIoT_proSeCap := ('1'B & ? length (7)) , 
    cIoT_V2X :=  (? length (2) & '0'B & ? length (5))
   };





After Change:
	template (present) UE_NetworkCap cdr_NR5GC_UENetworkCap_11_4_12 (template (omit) IEI8_Type p_IEI)
           modifies cr_UENetworkCap_Any :=
  {
    cIoT_proSeCap :=?,
    cIoT_V2X :=  (? length (2) & '0'B & ? length (5))
   };


2.7 Additional MCC160 implementation
	Function name
	f_TC_11_4_12_NR5GC_EUTRA()

	Reason for change
	1. Some co-ordination messages between NR and EUTRA are still missing in founction f_TC_11_4_12_NR5GC_EUTRA()

	Summary of change
	1. Add the necessary co-ordination messages in E-UTRA Preamble

	TTCN module
	Emergency_IRAT_NR5GC_EUTRA.ttcn



   function f_TC_11_4_12_NR5GC_EUTRA() runs on EUTRA_Multi5GS_PTC
  { //5GMM-REGISTERED.NORMAL-SERVICE / 5GMM-IDLE / Emergency call / Disabling N1 mode / Emergency call establishment over EPS / Success

    f_EUTRA38_IRAT_Init(c31, HANDLE_UM_DATA, USE_BIG_GRANTS);

    //Create and configure cell
    f_EUTRA_CellConfig_Def(eutra_Cell1);

    f_EUTRA38_NR5GC_PreambleAndSwitchOff(eutra_Cell1);
    f_EUTRA_NR_SendCoOrd (NR, cms_EUTRA_NR_Trigger("Preamble")); //Trigger to register on NR
    f_EUTRA_NR_WaitForCoOrd_Trigger (NR, "Preamble");            //Successfully registered on NR
    // Ok now registered on both RATs

    f_EUTRA_TestBody_Set(true);
    fl_TC_11_4_12_NR5GC_EUTRA_TestBody();
    f_EUTRA_TestBody_Set(false);

    f_EUTRA38_MultiPDN_Postamble(eutra_Cell1, E2_CONNECTED);
    f_EUTRA_NR_SendCoOrd (NR, cms_EUTRA_NR_Trigger("Postamble"));

  }

1. [bookmark: _Toc54888065][bookmark: _Toc112921502][bookmark: _Toc122434494][bookmark: _Toc295288971][bookmark: _Toc325725666]Branches executed
This NR5GC IRAT TC was executed in 5GS Stand Alone mode (Option 2) on NR Primary Band “n41” and EUTRA Primary Band e2 using NIA1 Integrity and NEA1 ciphering algorithms
4 [bookmark: _Toc112921503]Execution Log Files 
4.1 [bookmark: _Toc112921504]Qualcomm
The Qualcomm SM 8450,SDX65 UE  passed this test case on R&S® 5G Protocol Conformance Test platform and Keysight S8704A Protocol Conformance Toolset. The documentation below is enclosed as evidence of the successful test case run [1]:
Test case execution log file:
R&S :
tc_11_4_12_QC.log
(Note: PICS/PIXIT settings are captured at the beginning of the TLI log)
In the log file the complete test case execution as well as the PICS/PIXIT parameter settings can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.
Keysight :
TC_11_4_12_LOG.html.xml
TC_11_4_12_PIXIT.xml

5 [bookmark: _Toc122434496][bookmark: _Toc295288973][bookmark: _Toc325725668][bookmark: _Toc112921505]References

	R5s221059:   Supporting information for addition of NR5GC testcase 11.4.12 in FR1




